Упражнение 7

Темплейти. Ограничения в лявата част на правило

Конструкция deftemplate
Наредените факти задават информация позиционно. За да се достъпи тази информация потребителят трябвада знае  не само съхранените във факта данни, но и кое поле съдържа необходимите му данни. Ненаредените факти дават възможностна потребителя да придаде по-голяма степен на абстракция на факта чрез присвояване на име на всяко поле във факта. Конструкцията deftemplate се използва за създаване на темплейт, който след това може да се използва от ненаредени факти за достъпване на полета от факта по име. Тази конструкция е аналог на запис или структура в програмни езици като Паскал или С.

Синтаксисът на deftemplate е:

(deftemplate <deftemplate-name> [<comment>]

   <slot-definition>*)

<slot-definition> ::= <single-slot-definition> |

                      <multislot-definition>

<single-slot-definition> 

               ::= (slot <slot-name> 

                         <template-attribute>*)

<multislot-definition>   

               ::= (multislot <slot-name> 

                              <template-attribute>*)

<template-attribute> ::= <default-attribute> |

                         <constraint-attribute>

<default-attribute>  

               ::= (default ?DERIVE | ?NONE | <expression>*) |

                   (default-dynamic <expression>*)

Ако даден съществуващ темплейт се дефинира отново, новата дефиниция ще замести старата. Темплейтите не могатда се заместват с нови докато се използват (например от факт или офразец в правило). Темплейтът може да има произволен брой single или multifield слотове. CLIPS винаги изисква спазване на съответствие между единични или множествени слотове на темплейт и техните стойности. Например съхраняването (или съпоставянето) на множество стойности в single-field slot е грешно.

Пример:

(deftemplate object

   (slot name)

   (slot location)

   (slot on-top-of)

   (slot weight)

   (multislot contents))

Slot default стойности

<defaultattribute> определя стойността която да се използва за слотове на темплейтен факт без зададена стойност при изпълнение на командата assert. Могат да се изберат един или два типа на подразбиращи се селекции: default или dynamicdefault.

Атрибутите задават статично стойност по подразбиране. Определените изрази се изчисляват еднократно при дефинирането на темплейта и резултата се съхранява с deftemplate. Резултатът се присвоята на съответния слот при добавяне на нов темплейтен факт. Ако за подразбираща се стойност се използва ключовата дума ?DERIVE, то подразбиращата се стойност се определя от ограниченията за слота. По подразбиране атрибута default за слот е (default ?DERIVE). Ако за подразбираща се стойност се използва ключовата дума ?NONE, то стойността на слота трябва явно да се зададе при изпълнение на командата assert. Добавянето на темплейтен факт без задаване на стойности за (default ?NONE) слотове е грешно.

Атрибута defaultdynamic е динамична подразбираща се стойност. Определените изрази се изчисляват всеки път когато темплейтните факти се добавят и резултата се присвоята на съответния слот.

singlefield слот може да има само една подразбираща се стойност. За multifield слот може да се зададе произволен брой подразбиращи се стойности (докато броят на стойностите удовлетворява cardinality атрибута за слота).

Пример:

CLIPS> (clear)

CLIPS> 

(deftemplate foo

   (slot w (default ?NONE))

   (slot x (default ?DERIVE))

   (slot y (default (gensym*)))

   (slot z (default-dynamic (gensym*))))

CLIPS> (assert (foo))

[TMPLTRHS1] Slot w requires a value because of its (default ?NONE) attribute.

CLIPS> (assert (foo (w 3)))

<Fact-0>

CLIPS> (assert (foo (w 4)))

<Fact-1>

CLIPS> (facts)

f-0     (foo (w 3) (x nil) (y gen1) (z gen2))

f-1     (foo (w 4) (x nil) (y gen1) (z gen3))

For a total of 2 facts.

CLIPS>

Ограничение на слотове за стойности по подразбиране.

Singlefield слотове които не са зададени в образец в лявата част на правило съответстват на singlefield wildcard (?), а multifield слотовете на multifield wildcard ($?).

Атрибути за задаване на ограничения върху стойностите на слот.

За темплейтите се поддържат статични и динамични проверки за стойностите на единични и множествени слотове. Статичната проверка се изпълнява когато се проверяват (парсират) конструкциите или командите използващи темплейтните слотове (и при това определения темплейт асоцииран с конструкцията или командата може непосредствено да се провери). Темплейтните образци използвани в лявата част на правило също се проверяват за откриване на конфликти на ограниченията измежду променливите използвани в повече от един слот. Ако има грешки от използване на неправилни стойности това се съобщава веднага. Обръщенията към индекс на факт в команди като modify и duplicate се разглеждат като неопределени и не се проверяват от статичната проверка. Статичната проверка се изпълнява по подразбиране. Това може да се промени чрез използване на функцията setstaticconstraintchecking. Поддържа се и динамична проверка. Ако тя е разрешена, то стойностите на новите темплейтни факти се проверяват при добавянето им в списъка от факти. Динамичната проверка по подразбиране не се изпълнява. Това може да се промени, чрез използване на функцията setdynamicconstraintchecking. Ако се открие грешка при извършване на динамичната проверка, то изпълнението се прекъсва.

Пример:

(deftemplate object

   (slot name

      (type SYMBOL)

      (default ?DERIVE))

   (slot location

      (type SYMBOL)

      (default ?DERIVE))

   (slot on-top-of

      (type SYMBOL)

      (default floor))

   (slot weight

      (allowed-values light heavy)

      (default light))

   (multislot contents

      (type SYMBOL)

      (default ?DERIVE)))

Implied deftemplates

Добавянето или обръщението към наредени факти (като напр. образец в лявата част на правилата) създава implied deftemplate с единичен implied multifield слот. Неговото име не се отпечатва при отпечатването на факта. implied deftemplate може да се обработва и проверява подобно на всеки потребителски дефиниран deftemplate (въпреки че няма подходящ печатен формат).

Пример:

CLIPS> (clear)

CLIPS> (assert (foo 1 2 3))

<Fact-0>

CLIPS> (defrule yak (bar 4 5 6) =>)

CLIPS> (list-deftemplates)

initial-fact

foo

bar

For a total of 3 deftemplates.

CLIPS> (facts)

f-0     (foo 1 2 3)

For a total of 1 fact.

CLIPS>

Елемент на условието от тип образец

Един елемент на условието от тип образец с състои от множество от ограничения върху полетата, wildcards и променливи, които се използват за ограничаване на множеството от факти или инстанции които удовлетворяват елемента на условието. Елементите на условието се удовлетворяват от удовлетворяващи обекти (факти или инстанции на класове). Ограниченията върху полетата са множества от ограничения които се използват за проверка на едно поле или слот на удовлетворяващ обект. Едно ограничение върху поле може да съдържа само едно литерално ограничение, но може да съдържа и няколко свързани ограничения. Освен това CLIPS предоставя три други типа ограничения: ограничения от логически връзки, предикатни ограничения и ограничения от върнати стойности. Wildcards се използват в ограниченията от тип образец за задаване на съпоставимост на единично поле или група от полета с всякакъв друг обект. Променливите се използват за съхраняване на стойност на поле, за да може тя да се използва в LHS на правило в друг елемент на условието или в RHS на правилото като агрумент при извършване на определена операция.

Първото поле на всеки образец трябва да е symbol и не може да използва някакви други ограничения. Това първо поле се използва от CLIPS за определяне дали образеца съответства на нареден факт, темплейтен факт или на инстанция. Символът object е резервиран за образец обект. Всеки друг използван symbol трябва да отговаря на име на deftemplate (или ще бъде създаден косвен deftemplate). Имената на слотове също трябва да са symbols и не могат да съдържат други ограничения.

За образци обекти и темплейти, един слот може да съдържа само едно ограничение и това ограничение трябва да може да се удовлетвори от едно поле (без множествени wildcards или променливи). Един множествен слот може да съдържа произволен брой ограничения върху полета.

Примерите и синтаксиса в следващите секции са за наредени факти, темплейти и обекти. Използват се следните конструкции:

(deffacts data-facts

(data 1.0 blue "red")

(data 1 blue)

(data 1 blue red)

(data 1 blue RED)

(data 1 blue red 6.9))

(deftemplate person

(slot name)

(slot age)

(multislot friends))

(deffacts people

(person (name Joe) (age 20))

(person (name Bob) (age 20))

(person (name Joe) (age 34))

(person (name Sue) (age 34))

(person (name Sue) (age 20)))

Литерални ограничения

Най-простите ограничения които могат да се използват в елемент на условието от тип образец са тези които задавт точна стойност която трябва да съвпада с дадено поле. Тя се нарича литерално ограничения. Едно литерално ограничение се състои изцяло от константи като реални числа, цели числа, символи, низове и имена на инстанции. То не съдържа променливи или wildcards. Всички ограничения в литералното ограничения трябва да съвпадат точно с всички полета на удовлетворяващия обект.

Синтаксис

Един условен елемент от вид нареден образец състоящ се само от литерали има следния основен синтаксис:

(<constant-1> ... <constant-n>)

Елемент на условието от тип темплейт съдържащ само литерали има следния синтаксис:

(<deftemplate-name> (<slot-name-1> <constant-1>)

•

•

•

(<slot-name-n> <constant-n>))

Пример 1

Този пример използва deffacts data-facts от предишните параграфи

CLIPS> (clear)

CLIPS> (defrule find-data (data 1 blue red) =>)

CLIPS> (reset)

CLIPS> (agenda)

0 find-data: f-3

For a total of 1 activation.

CLIPS> (facts)

f-0 (initial-fact)

f-1 (data 1.0 blue "red")

f-2 (data 1 blue)

f-3 (data 1 blue red)

f-4 (data 1 blue RED)

f-5 (data 1 blue red 6.9)

For a total of 6 facts.

CLIPS>

Пример 2

Този пимер използва темплейта person и deffacts people от предишните параграфи

CLIPS> (clear)

CLIPS>

(defrule Find-Bob

(person (name Bob) (age 20))

=>)

CLIPS>

(defrule Find-Sue

(person (age 34) (name Sue))

=>)

CLIPS> (reset)

CLIPS> (agenda)

0 Find-Sue: f-4

0 Find-Bob: f-2

For a total of 2 activations.

CLIPS> (facts)

f-0 (initial-fact)

f-1 (person (name Joe) (age 20) (friends))

f-2 (person (name Bob) (age 20) (friends))

f-3 (person (name Joe) (age 34) (friends))

f-4 (person (name Sue) (age 34) (friends))

f-5 (person (name Sue) (age 20) (friends))

For a total of 6 facts.

CLIPS>

Единични и множествени wildcards

CLIPS има два wildcard символа които могат да се използват в образец за съпоставяне с полета. CLIPS интерпретира тези wildcard символи като намиращи се на определено място в някаква част от удовлетворяващ обект. Единичния wildcard означен с въпросителен знак (?) се съпоставя успешно с всяка друга стойност намираща се в точно едно поле на удовлетворяващия факт. Множествения wildcard, означенс доларов знак следван от въпросителен знак ($?) се съпоставя успешно с всяка стойност в нула или повече полета в удовлетворяващ факт. Единичните и множествени wildcard могат да се комбинират в един образец при което се получават всички възможни комбинации. Не е възможно използването на множетсвен wildcard на мястото на единичен слот в темплейтен или обектен образец. По подразбиране, незададен единичен слот в темплейтен/обектен образец се удовлетворява от косвен множествен wildcard.

Синтаксис

Условен елемент от вид нареден образец съдържащ само литерали и wildcards има следния синтаксис:

(<constraint-1> ... <constraint-n>)

където

<constraint> ::= <constant> | ? | $?

Условен елемент от тип темплейтен образец съдържащ само литерали и wildcards има следния синтаксис:

(<deftemplate-name> (<slot-name-1> <constraint-1>)

•

•

•

(<slot-name-n> <constraint-n>))

Пример 1

Този пример използва deffacts data-facts от предишните параграфи

CLIPS> (clear)

CLIPS>

(defrule find-data

(data ? blue red $?)

=>)

CLIPS> (reset)

CLIPS> (agenda)

0 find-data: f-5

0 find-data: f-3

For a total of 2 activations.

CLIPS> (facts)

f-0 (initial-fact)

f-1 (data 1.0 blue "red")

f-2 (data 1 blue)

f-3 (data 1 blue red)

f-4 (data 1 blue RED)

f-5 (data 1 blue red 6.9)

For a total of 6 facts.

CLIPS>

Пример 2

Този пимер използва темплейта person и deffacts people от предишните параграфи

CLIPS> (clear)

CLIPS>

(defrule match-all-persons

(person)

=>)

CLIPS> (reset)

CLIPS> (agenda)

0 match-all-persons: f-5

0 match-all-persons: f-4

0 match-all-persons: f-3

0 match-all-persons: f-2

0 match-all-persons: f-1

For a total of 5 activations.

CLIPS> (facts)

f-0 (initial-fact)

f-1 (person (name Joe) (age 20) (friends))

f-2 (person (name Bob) (age 20) (friends))

f-3 (person (name Joe) (age 34) (friends))

f-4 (person (name Sue) (age 34) (friends))

f-5 (person (name Sue) (age 20) (friends))

For a total of 6 facts.

CLIPS>

Множетсвен wildcard и литерални ограничения могат да се комбинират за съставяне на мощни елементи на условия. Един образец който търси удовлетворяващ обект, който трябва да съдържа символа YELLOW във всяко поле (с изключение на първото) може да се запише по следния начин:

(data $? YELLOW $?)

Някой примери за успешно съпоставяне с този образец са:

(data YELLOW blue red green)

(data YELLOW red)

(data red YELLOW)

(data YELLOW)

(data YELLOW data YELLOW)

Последният факт ще удовлетвори два пъти образеца, тъй като YELLOW се съдържа два пъти във факта. Използването на множествен wildcard трябва да се ограничава до случаи на образци при които единичния wildcard не може да създаде образец които да удовлетворява необходимото съпоставяне, тъй като множествения wildcard извършва всички възможни комбинации на съпоставяне които могат да се получат от удовлетворяващия обект. Това комбиниране на съпоставяния изисква значително количество време за изпълнение в сравнение с времето за изпълнение на единично съпоставяне.

Единични и множествени променливи

Wildcard символите заместват части от образците и приемат всяка стойност. Стойността на полето което се замества може да се получи в променлива за сравняване, извеждане на екран или други манипулации. Това се извършва, чрез директно използване след wildcard на име на променлива.

Синтаксис

Разширявайки дефиницията на синтаксиса даден в предишните параграфи се получава:

<constraint> ::= <constant> | ? | $? |

<single-field-variable> |

<multifield-variable>

<single-field-variable> ::= ?<variable-symbol>

<multifield-variable> ::= $?<variable-symbol>

където <variable-symbol> е подобен на symbol с тази разлика, че трябва да започва с буква. Двойните кавички не са разрешени като част от име на променлива. Правилата за съпоставяне на образци са подобни на тези за wildcard символи. При първата поява, променливата действа точно като wildcard по това, че тя се свързва с всяка стойност в поле(та). След това обаче, използването на променлива изисква съответното поле да съответства на свързаната променлива. Свързването е истинно само в рамките на правилото в което се среща. Всяко правило имачастен списък от имена на променливи с техни асоциирани стойности. Така променливите са локални за правилото. Свързани променливи могат да се подават към външни функции. Операторът $ има специално значение в LHS като оператор за съпоставяне за индикиране на нула или повече полета които трябва да се съпоставят. В други места (като RHS на правило), символът $ пред променлива означава че трябва да се извърши разширяване на последователността преди извикването на функцията. Така когато са подадени като параметри при функционално извикване (или в LHS или в RHS на правило), множествените променливи не трябва да се предшестват от $ (освен ако не е необходимо разширяване на последователността). Всички други използвания на множествена променлива в LHS на правило трябва да използват $. Не е разрешено използването на множествена променлива в единичен слот на темплейтен/обектен образец.

Пример 1

CLIPS> (clear)

CLIPS> (reset)

CLIPS> (assert (data 2 blue green)

(data 1 blue)

(data 1 blue red))

<Fact-3>

CLIPS> (facts)

f-0 (initial-fact)

f-1 (data 2 blue green)

f-2 (data 1 blue)

f-3 (data 1 blue red)

For a total of 4 facts.

CLIPS>

(defrule find-data-1

(data ?x ?y ?z)

=>

(printout t ?x " : " ?y " : " ?z crlf))

CLIPS> (run)

1 : blue : red

2 : blue : green

CLIPS>

Пример 2

CLIPS> (reset)

CLIPS> (assert (data 1 blue)

(data 1 blue red)

(data 1 blue red 6.9))

<Fact-3>

CLIPS> (facts)

f-0 (initial-fact)

f-1 (data 1 blue)

f-2 (data 1 blue red)

f-3 (data 1 blue red 6.9)

For a total of 4 facts.

CLIPS>

(defrule find-data-1

(data ?x $?y ?z)

=>

(printout t "?x = " ?x crlf

"?y = " ?y crlf

"?z = " ?z crlf

"------" crlf))

CLIPS> (run)

?x = 1

?y = (blue red)

?z = 6.9

------

?x = 1

?y = (blue)

?z = red

------

?x = 1

?y = ()

?z = blue

------

CLIPS>

Веднъж след като е извършено свързване на променлива, всички нейни изполвания за съпоставяне съответстват на стойността от първото и свързване. Това се извършва и при единични и при множествени променливи. Същото важи и когато се извършва между образци.

Пример 3

CLIPS> (clear)

CLIPS>

(deffacts data

(data red green)

(data purple blue)

(data purple green)

(data red blue green)

(data purple blue green)

(data purple blue brown))

CLIPS>

(defrule find-data-1

(data red ?x)

(data purple ?x)

=>)

CLIPS>

(defrule find-data-2

(data red $?x)

(data purple $?x)

=>)

CLIPS> (reset)

CLIPS> (facts)

f-0 (initial-fact)

f-1 (data red green)

f-2 (data purple blue)

f-3 (data purple green)

f-4 (data red blue green)

f-5 (data purple blue green)

f-6 (data purple blue brown)

For a total of 7 facts.

CLIPS> (agenda)

0 find-data-2: f-4,f-5

0 find-data-1: f-1,f-3

0 find-data-2: f-1,f-3

For a total of 3 activations.

CLIPS>

Свързващи ограничения

Могат да се използват три свързващи ограничения за свързване на отделни ограничения и променливи едно с друго. Те са & (И), | (ИЛИ) и ~ (НЕ) свързващите ограничения. Ограничението & се удовлетворява ако и двете свързващи се ограничения са удвлетворени. Ограничението | се удовлетворява ако кое да е от двете свързващи се ограничения се удовлетворява. Ограничението ~ се удовлетворява ако последващото ограничение не се удовлетворява. Свързващите ограничения могат да се комбинират по почти всеки възможен начин или брой за да се ограничи стойността на съответното поле в образеца. Ограничението ~ има най-висок приоритет, следван от ограничението &, следвано от |. Оценяването на множество ограничения се извършва отляво надясно. Има едно изключение към правилата за приоритет които се прилагат към свързване на променливи. Ако първото ограничение е променлива следвана от свързващото ограничение &, то първото ограничение се разглежда като отделно ограничение , което също трябва да се удовлетворява. Така ограничението ?x&red|blue се разглежда като ?x&(red|blue), а не като (?x&red)|blue както се получава според нормалните правила за приоритет.

Основен синтаксис

Свързващите ограничения имат следния синтаксис:

<term-1>&<term-2> ... &<term-3>

<term-1>|<term-2> ... |<term-3>

~<term>

където <term> може да е единина променлива, множествена променлива, константа или свързващо ограничение.

Синтаксис

Разширявайки синтаксиса от дефиницията от предишните параграфи се получава:

<constraint> ::= ? | $? | <connected-constraint>

<connected-constraint>

::= <single-constraint> |

<single-constraint> & <connected-constraint> |

<single-constraint> | <connected-constraint>

<single-constraint> ::= <term> | ~<term>

<term> ::= <constant> |

<single-field-variable> |

<multifield-variable>

Ограничението & обикновено се използва само за конюнкция с други ограничения или свързвания на променливи. Забележете, че свързващите ограничения могат да се използват заедно със свързвания на променливи. Ако първият член на свързващо ограничение е първо срещане на име на променлива, то полето се бъде ограничено само от оставащите ограничения за полето. Променливата ще бъде свързана със стойността на полето. Ако променливата преди това е била свързана, то тя се разглежда като допълнително ограничение заедно с оставащите ограничения за полето, т.е. полето трябва да има същата стойност към която вече е свързана променливата и трябва да удовлетворява ограниченията върху полето.

Пример 1

CLIPS> (clear)

CLIPS> (deftemplate data-B (slot value))

CLIPS>

(deffacts AB

(data-A green)

(data-A blue)

(data-B (value red))

(data-B (value blue)))

CLIPS>

(defrule example1-1

(data-A ~blue)

=>)

CLIPS>

(defrule example1-2

(data-B (value ~red&~green))

=>)

CLIPS>

(defrule example1-3

(data-B (value green|red))

=>)

CLIPS> (reset)

CLIPS> (facts)

f-0 (initial-fact)

f-1 (data-A green)

f-2 (data-A blue)

f-3 (data-B (value red))

f-4 (data-B (value blue))

For a total of 5 facts.

CLIPS> (agenda)

0 example1-2: f-4

0 example1-3: f-3

0 example1-1: f-1

For a total of 3 activations.

CLIPS>

Пример 2

CLIPS> (clear)

CLIPS> (deftemplate data-B (slot value))

CLIPS>

(deffacts B

(data-B (value red))

(data-B (value blue)))

CLIPS>

(defrule example2-1

(data-B (value ?x&~red&~green))

=>

(printout t "?x in example2-1 = " ?x crlf))

CLIPS>

(defrule example2-2

(data-B (value ?x&green|red))

=>

(printout t "?x in example2-2 = " ?x crlf))

CLIPS> (reset)

CLIPS> (run)

?x in example2-1 = blue

?x in example2-2 = red

CLIPS>

Пример 3

CLIPS> (clear)

CLIPS> (deftemplate data-B (slot value))

CLIPS>

(deffacts AB

(data-A green)

(data-A blue)

(data-B (value red))

(data-B (value blue)))

CLIPS>

(defrule example3-1

(data-A ?x&~green)

(data-B (value ?y&~?x))

=>)

CLIPS>

(defrule example3-2

(data-A ?x)

(data-B (value ?x&green|blue))

=>)

CLIPS>

(defrule example3-3

(data-A ?x)

(data-B (value ?y&blue|?x))

=>)

CLIPS> (reset)

CLIPS> (facts)

f-0 (initial-fact)

f-1 (data-A green)

f-2 (data-A blue)

f-3 (data-B (value red))

f-4 (data-B (value blue))

For a total of 5 facts.

CLIPS> (agenda)

0 example3-3: f-1,f-4

0 example3-3: f-2,f-4

0 example3-2: f-2,f-4

0 example3-1: f-2,f-3

For a total of 4 activations.

CLIPS>

Предикатни ограничения

Понякога е необходимо да се ограничи едно поле на базата на истинността на определен логически израз. CLIPS дава възможност за използване на предикатно ограничение за ограничаване на поле по този начин. Предикатното ограничение дава възможност да се извика предикатна функция (връщаща символ FALSE при неудовлетворяване и стойност различна от FALSE при удовлетворяване) по време на процеса на съпоставяне. Ако предикатната функция връща стойност различна от FALSE, ограничението се удовлетворява. Ако предикатната функция връща символ FALSE, ограничението не се удовлетворява. Предикатното ограничение се извиква след двоеточие, чрез подходящо функционално извикване към предикатна функция. Обикновено, предикатните ограничения се използват заедно със свързващи ограничения и свързване на променлива (т.е. променливата трябва да се свърже със стойност и след това да се използва в предикатно ограничение).

Основен синтаксис

:<function-call>

Синтаксис

Разширявайки дефиницията на синтаксиса в предишните параграфи се получава:

<term> ::= <constant> |

<single-field-variable> |

<multifield-variable> |

:<function-call>

Могат да се използват множество предикатни ограничения за ограничаване на едно поле. CLIPS предоставя няколко предикатни функции. Потербителя също може да разработи свои собствени предикатни функции.

Пример 1:

CLIPS> (clear)

CLIPS>

(defrule example-1

(data ?x&:(numberp ?x))

=>)

CLIPS> (assert (data 1) (data 2) (data red))

<Fact-2>

CLIPS> (agenda)

0 example-1: f-1

0 example-1: f-0

For a total of 2 activations.

CLIPS>

Пример 2:

CLIPS> (clear)

CLIPS>

(defrule example-2

(data ?x&~:(symbolp ?x))

=>)

CLIPS> (assert (data 1) (data 2) (data red))

<Fact-2>

CLIPS> (agenda)

0 example-2: f-1

0 example-2: f-0

For a total of 2 activations.

CLIPS>

Пример 3:

CLIPS> (clear)

CLIPS>

(defrule example-3

(data ?x&:(numberp ?x)&:(oddp ?x))

=>)

CLIPS> (assert (data 1) (data 2) (data red))

<Fact-2>

CLIPS> (agenda)

0 example-3: f-0

For a total of 1 activation.

CLIPS>

Пример 4:

CLIPS> (clear)

CLIPS>

(defrule example-4

(data ?y)

(data ?x&:(> ?x ?y))

=>)

CLIPS> (assert (data 3) ; f-0

(data 5) ; f-1

(data 9)) ; f-2

<Fact-2>

CLIPS> (agenda)

0 example-4: f-0,f-2

0 example-4: f-1,f-2

0 example-4: f-0,f-1

For a total of 3 activations.

CLIPS>

Пример 5:

CLIPS> (clear)

CLIPS>

(defrule example-5

(data $?x&:(> (length$ ?x) 2))

=>)

CLIPS> (assert (data 1) ; f-0

(data 1 2) ; f-1

(data 1 2 3)) ; f-2

<Fact-2>

CLIPS> (agenda)

0 example-5: f-2

For a total of 1 activation.

CLIPS>

Ограничение от тип върната стойност

Може да се използва върната стойност от външна функция за ограничаване на стойност на поле. Ограничението от тип върната стойност (=) дава възможност на потребителя да извършва обръщения към външни функции от вътрешността на образец. (Това ограничение е различно от функцията за сравнение която използва същия символ. Върнатата стойност трябва да бъде от примитивен тип. Тази стойност се включва директно в образеца на позицията в която функцията е извикана все едно на нейното място се намира литерално ограничение и всеки удовлетворяващ обект трябва да удовлетворява тази стойност по същия начин както ако в правилото е записана стойността. Важно е да се обърне внимание на това, че функцията се оценява всеки път когато се проверява ограничението (а не само веднъж).

Основен синтаксис

=<function-call>

Синтаксис

Разширявайки дефиницията на синтаксиса от предишните параграфи се получава:

<term> ::= <constant> |

<single-field-variable> |

<multifield-variable> |

:<function-call> |

=<function-call>

Пример 1:

CLIPS> (clear)

CLIPS> (deftemplate data (slot x) (slot y))

CLIPS>

(defrule twice

(data (x ?x) (y =(* 2 ?x)))

=>)

CLIPS> (assert (data (x 2) (y 4)) ; f-0

(data (x 3) (y 9))) ; f-1

<Fact-1>

CLIPS> (agenda)

0 twice: f-0

For a total of 1 activation.

CLIPS>

Пример 2:

CLIPS> (clear)

CLIPS>

(defclass DATA (is-a USER)

(role concrete) (pattern-match reactive)

(slot x (create-accessor write)))

CLIPS>

(defrule return-value-example-2

(object (is-a DATA)

(x ?x1))

(object (is-a DATA)

(x ?x2&=(+ 5 ?x1)|=(- 12 ?x1)))

=>)

CLIPS> (make-instance of DATA (x 4))

[gen1]

CLIPS> (make-instance of DATA (x 9))

[gen2]

CLIPS> (make-instance of DATA (x 3))

[gen3]

CLIPS> (agenda)

0 return-value-example-2: [gen3],[gen2]

0 return-value-example-2: [gen2],[gen3]

0 return-value-example-2: [gen1],[gen2]

For a total of 3 activations.

CLIPS>

Удовлетворяване на образци с обектни образци

Инстанции на потребителски дефинирани класове могат да бъдат съпоставяни в лявата част на правилата. Образците могат да се съпоставят само с обекти за които най-специфичния клас е дефиниран преди образеца и които са в обхвата на текущия модул. Класове които могат да имат обекти които се съпоставят с образци не могат да се изтрият или променят докато не се премахне образеца. Дори ако правило се изтрие от неговата RHS, класовете свързани с неговите  образци не могат да се променят освен след края на изпълнението на RHS.

Когато се създаде или премахне инстанция, всички образци съпоставими към този обект се обновяват. Обаче ако се промени някой слот, това се отразява само на тези образци които се съпоставят явно със слота. Така може да се използват логически зависимости за да се обвърже промяна с определен слот.

Промени на нереактивни слотове или инстанции на нереактивни класове не оказват влияние върху класовете. Освен това мрежата Rete не е съвсем ясна веднага след промени на определен слот ако съпоставянето се забави, чрез използване на функциите make-instance, initialize-instance, modify-instance, message-modify-instance, duplicate-instance, message-duplicate-instance или object-match-delay.

Синтаксис

<object-pattern> ::= (object <attribute-constraint>*)

<attribute-constraint> ::= (is-a <constraint>) |

(name <constraint>) |

(<slot-name> <constraint>*)

Това is-a ограничение се използва за специфициране на ограничение което съответства на въпроса "Този обект инстанция ли е на класа FOO?". Ограничението is-a обхваща и подкласове на удовлетворяващите класове освен ако не е указано друго в образеца. Ограничението name се използва за задаване на определена инстанция с която да се извършва съпоставяне. Оценката на името трябва да е от примитивния тип instance-name, а не symbol. Multifield ограниченията (като $?) не могат да се използват с is-a или name ограниченията. При всички други ограничения с изключение на гореспоменатите специални случаи, са подобни на ограниченията при темплейтни слотове. Както при темплейтни образци, имената на слотове за обектни образци трябва да са symbols и не могат да съдържат някакви други ограничения.

Пример 1

Следните правила илюстрират удовлетворяването на образци за обектен клас

(defrule class-match-1

(object)

=>)

(defrule class-match-2

(object (is-a FOO))

=>)

(defrule class-match-3

(object (is-a FOO | BAR))

=>)

(defrule class-match-4

(object (is-a ?x))

(object (is-a ~?x))

=>)

Правилото class-match-1 се удовлетворява от всички инстанции на всеки реактивен клас. Правилото class-match-2 се удовлетворява от всички инстанции на класа FOO. Правилото class-match-3 се удовлетворява от всички инстанции на класовете FOO или BAR. Правилото class-match-4 ще се удовлетвори от кои да е две инстанции на взаимно изключваши се класове.

Пример 2

Следните правила илюстрират удовлетворяването на образци за различни атрибути на слотове на обекти.

(defrule slot-match-1

(object (width))

=>)

(defrule slot-match-2

(object (width ?))

=>)

(defrule slot-match-3

(object (width $?))

=>)

Правилото slot-match-1 се удовлетворява от всички инстанции на реактивни класове които съдържат реактивен слот width с множествена стойност с нулева дължина. Правилото slot-match2 се удовлетворява от всики инстанции на реактивни класове, които съдържат единичен или множествен слот width свързан с единична стойност. Правилото slot-match-3 се удовлетворява от всички инстанции на реактивни класове които съдържат реактивен единичен или множествен слот width свързан с какво да е число или стойности. Слот съдържащ множествена стойност с нулева дължина ще удовлетвори правилата slot-match-1 и slot-match-3, но няма да удовлетвори правило slot-match-2 (тъй като кардиналността на стойността е нула).

Пример 3

Следните правила илюстрират удовлетворяването на образци за стойности на слотове на обект

(defrule value-match-1

(object (width 10)

=>)

(defrule value-match-2

(object (width ?x&:(> ?x 20)))

=>)

(defrule value-match-3

(object (width ?x) (height ?x))

=>)

Правилото value-match-1 се удовлетворява от всички инстанции на реактивни класове, които съдържат реактивен слот width със стойност 10. Правилото value-match-2 се удовлетворява от всички инстанции на реактивни класове, които съдържат реактивен слот width, който има стойност по-голяма от 20. Правилото value-match-3 се удовлетворява от всички инстанции на реактивни класове, които съдържат реактивни слотове width и height с една и съща стойност.

Адреси на образци

Определени действия в дясната част на правилата като retract и unmake-instance извършват действия върху целия образец в узловието. За да се обозначи за кой факт или инстанция те действат, може да се свърже променлива с адрес на факт или инстанция (fact-address или instance-address) на образец в условието. Адреси на факт и инстанция се наричат под общото наименование адрес на образец.

Синтаксис:

<assigned-pattern-CE> ::= ?<variable-symbol> <- <pattern-CE>

Лявата стрелка <- е част от синтаксиса. Променливата свързана с адрес на факт или инстанция може да се сравнява с друи променливи или да се използва при обръщение към външни функции. Променливите свързани с адрес на факт или инстанция могат да се използват след това за задаване на ограничения върху полета в образец на условието, като обратното не е възможно. Грешно е да се свърже променлива към елемент на условието not.

Примери:

(defrule dummy

(data 1)

?fact <- (dummy pattern)

=>

(retract ?fact))

(defrule compare-facts-1

?f1 <- (color ~red)

?f2 <- (color ~green)

(test (neq ?f1 ?f2))

=>

(printout t "Rule fires from different facts" crlf))

(defrule compare-facts-2

?f1 <- (color ~red)

?f2 <- (color ~green&:(neq ?f1 ?f2))

=>

(printout t "Rule fires from different facts" crlf))

(defrule print-and-delete-all-objects

?ins <- (object)

=>

(send ?ins print)

(unmake-instance ?ins))

Елемент на условието test

Ограниченията върху полетата използвани при образците дават възможност за задаване на много детайлни ограничения при удовлетворяване. Някой допълнителни възможности се предоставят, чрез елемента на условието от тип test. Условният елемент test се удовлетворява ако функционалното извикване в test се оценява със стойност различна от FALSE. Както при предикатните ограничения, потребителя може да сравнява стойности вече свързани с някакви променливи. По този начин могат да се извършват сложни математически и логически сравнения върху променливи (напр. разликата между ?x и ?y по-голяма ли е от определена стойност?). В тази конструкция могат да се използват и външни функции за сравняване на променливи.

В ограничението test могат да се използват всякакви външни функции. Потребителски дефинираните предикатни функции трябва да приемат аргументи по дефиниция (Advanced Programming Guide). В CLIPS има вградени някой предикатни функции.

Синтаксис:

<test-CE> ::= (test <function-call>)

Правилата не могат да използват test като първи елемент на условието. Той се оценява ако всички предходни елементи на условието са удовлетворени. Това означава, че test ще се оцени повече от веднъж ако предходните елементи на условието могат да се удовлетворят от повече от една група от образци. За да се предизвика повторно оценяване на елемента test, обекта удовлетворяващ елемента на условието преди test трябва да се промени. При използването на елемент на условието test може да се наложи добавяне на допълнителни елементи на условието в лявата част на правилото. Освен това може да се извърши автоматично преподреждане на test от CLIPS.

Пример 1

Този пример проверява дали разликата между две числа е по-голяма или равна на три:

CLIPS> (clear)

CLIPS>

(defrule example-1

(data ?x)

(value ?y)

(test (>= (abs (- ?y ?x)) 3))

=>)

CLIPS> (assert (data 6) (value 9))

<Fact-1>

CLIPS> (agenda)

0 example-1: f-0,f-1

For a total of 1 activation.

CLIPS>

Пример 2:

Този пример проверява дали наклона на линията свързваща две точки е положителен

CLIPS> (clear)

CLIPS>

(deffunction positive-slope

(?x1 ?y1 ?x2 ?y2)

(< 0 (/ (- ?y2 ?y1) (- ?x2 ?x1))))

CLIPS>

(defrule example-2

(point ?a ?x1 ?y1)

(point ?b ?x2 ?y2)

(test (> ?b ?a))

(test (positive-slope ?x1 ?y1 ?x2 ?y2))

=>)

CLIPS>

(assert (point 1 4.0 7.0) (point 2 5.0 9.0))

<Fact-1>

CLIPS> (agenda)

0 example-2: f-0,f-1

For a total of 1 activation.

CLIPS>

Важна забележка:

Тъй като test може да предизвика добавяне на образец initial-fact или initial-object към правилото, трябва да се изпълни командата reset (която създава факт initial-fact и инстанция initial-object) за да може test да работи правилно при всички случаи.

Елемент на условието Or (или)

Елементът на условието or дава възможност кой да е от няколко елемента да активира правилото. Ако някой от условните елементи в or е изпълнено, то or се удовлетворява. Ако всички други елементи на условието в LHS се удовлетворяват, правилото ще се активира. При това правилото ще се активира за всеки условен елемент с or който е удовлетворен (при предположение, че другите елементи на условието в правилото също се удовлетворяват). В or  могат да се намират произволен брой условни елементи. Ефекта от използването на този тип условие е идентичен както ако има няколко правила с подобни леви части и десни части.

Синтаксис

<or-CE> ::= (or <conditional-element>+)

Ако повече от един от условните елементи в or се удовлетворяват, правилото ще сработи толкова пъти, колкото удовлетворени комбинации от условия могат да се съставят.

Пример:

(defrule system-fault

(error-status unknown)

(or (temp high)

(valve broken)

(pump (status off)))

=>

(printout t "The system has a fault." crlf))

Този пример е еквивалентен на следните три (отделни) правила:

(defrule system-fault

(error-status unknown)

(pump (status off))

=>

(printout t "The system has a fault." crlf))

(defrule system-fault

(error-status unknown)

(valve broken)

=>

(printout t "The system has a fault." crlf))

(defrule system-fault

(error-status unknown)

(temp high)

=>

(printout t "The system has a fault." crlf))

Елемент на условието and

В CLIPS се приема, че всички правила имат неявен and елемент обхващащ условните елементи на LHS. Това означава, че всички елементи на условията в LHS трябва да се удовлетворят преди правилото да се активира. Явния елемент and на условието дава възможност за смесено използване на and и or елементи на условието. Така могат да се групират и други типове елементи на условието заедно с or и not. Елементът на условието and се удовлетворява, ако всички условни елементи в явния and  са удовлетворени. Ако всички други условия в LHS са истина, правилото ще се активира. В and  могат да се намират произволен брой условни елементи.

Синтаксис

<and-CE> ::= (and <conditional-element>+)

Пример:

(defrule system-flow

(error-status confirmed)

(or (and (temp high)

(valve closed))

(and (temp low)

(valve open)))

=>

(printout t "The system is having a flow problem." crlf))

Условен елемент and, който съдържа test или not като първи елемент на условието има образец (initial-fact) или (initial-object) добавен към първия елемент. LHS на всяко правило е обхванато от вграден and. Например следното правило:

(defrule nothing-to-schedule

(not (schedule ?))

=>

(printout t "Nothing to schedule." crlf))

се преобразува до:

(defrule nothing-to-schedule

(and (initial-fact)

(not (schedule ?)))

=>

(printout t "Nothing to schedule." crlf))

Елемент на условието not

Понякога отсъствието на информация също има смисъл, напр. едно правило може да се накара да сработи ако даден факт или друг елемент не съществува. Елементът на условието not се използва за тази цел. Той се удовлетворява ако условният елемент намиращ се в него не се удовлетворява. Както и при другите елементи на условието, в лявата част на правилото може да има всякакви други елементи на условието и в образеца за отрицание може да се използва някое константно поле взето от тях.

Синтаксис

<not-CE> ::= (not <conditional-element>)

В даден момент само един елемент на условието може да се използва в отрицание. Повече образци могат да се използват в отрицание, чрез използване на няколко not елемента. Трябва да се внимава при комбинирането на not с or и and, тъй като резултатите не винаги са очевидни. Същото важи и за свързвания на променливи в елемент на условието not. Свързани вече променливи могат свободно да се използват в not. Ако обаче променливите се свързват в елемента not, те могат да се използват само в него.

Примери:

(defrule high-flow-rate

(temp high)

(valve open)

(not (error-status confirmed))

=>

(printout t "Recommend closing of valve due to high temp"

crlf))

(defrule check-valve

(check-status ?valve)

(not (valve-broken ?valve))

=>

(printout t "Device " ?valve " is OK" crlf))

(defrule double-pattern

(data red)

(not (data red ?x ?x))

=>

(printout t "No patterns with red green green!" crlf ))

Елемент на условието not, който съдържа test се конвертира така, че test да се съдържа в and елемент и се предшества от образец (initial-fact) или (initial-object).

(not (test (> ?time-1 ?time-2)))

се конвертира до

(not (and (initial-fact)

(test (> ?time-1 ?time-2))))

По-лесно е просто да се конвертира test елемента по следния начин:

(test (not (> ?time-1 ?time-2)))

Важна забележка

Тъй като елементът на условието not може да предизвика добавяне на образец на факт (initial-fact) или образец на инстанция (instance-object), трябва да се изпълни команда reset.

Елемент на условието exists

Елементът на условието exists препоставя възможност за определяне дали група от други елементи на условие се удовлетворяват от поне едно множество от удовлетворяващи обекти (факти).

Синтаксис

<exists-CE> ::= (exists <conditional-element>+)

Елементът exists е реализиран, чрез два вложени not елемента. Например следното правило

(defrule example

(exists (a ?x) (b ?x))

=>)

е еквивалентно на правилото по-долу

(defrule example

(not (not (and (a ?x) (b ?x))))

=>)

Поради реализация на exists с елементи not, ограниченията към условните елементи в not (като свързване на адрес на факт) се прилагат и към exists.

Пример

CLIPS> (clear)

CLIPS>

(deftemplate hero

(multislot name)

(slot status (default unoccupied)))

CLIPS>

(deffacts goal-and-heroes

(goal save-the-day)

(hero (name Death Defying Man))

(hero (name Stupendous Man))

(hero (name Incredible Man)))

CLIPS>

(defrule save-the-day

(goal save-the-day)

(exists (hero (status unoccupied)))

=>

(printout t "The day is saved." crlf))

CLIPS>

Следните команди показват, че въпреки че има три факта, които удовлетворяват втория елемент на условието в правилото save-the-day, удовлетворяването е само едно.

CLIPS> (reset)

CLIPS> (agenda)

0 save-the-day: f-1,

For a total of 1 activation.

CLIPS> (facts)

f-0 (initial-fact)

f-1 (goal save-the-day)

f-2 (hero (name Death Defying Man) (status unoccupied))

f-3 (hero (name Stupendous Man) (status unoccupied))

f-4 (hero (name Incredible Man) (status unoccupied))

For a total of 5 facts.

CLIPS> (matches save-the-day)

Matches for Pattern 1

f-1

Matches for Pattern 2

f-0

Matches for Pattern 3

f-2

f-3

f-4

Partial matches for CEs 1 - 2

f-1,

Activations

f-1,

CLIPS>

Важна забележка

Елементът exists е реализиран, чрез елемент not. Тъй като елементът на условието not може да предизвика добавяне на образец на факт (initial-fact) или образец на инстанция (instance-object), трябва да се изпълни команда reset.

Елемент на условието forall

Елементът на условието forall е реализиран, чрез комбинация от елементи not и and. Например следното правило:

(defrule example

(forall (a ?x) (b ?x) (c ?x))

=>)

е еквивалентно на правилото:

(defrule example

(not (and (a ?x)

(not (and (b ?x) (c ?x)))))

=>)

Поради реализация на forall с елементи not, ограниченията към условните елементи в not (като свързване на адрес на факт) се прилагат и към forall.

Пример:

Следното правило определя, чрез forall дали всеки ученик е преминал успешно изпитите четене, писане и смятане:

CLIPS> (clear)

CLIPS>

(defrule all-students-passed

(forall (student ?name)

(reading ?name)

(writing ?name)

(arithmetic ?name))

=>

(printout t "All students passed." crlf))

CLIPS>

Следните команди илюстрират как работи forall в правилото all-students-passed. Първоначално това правило се удовлетворява, тъй като няма добавени ученици.

CLIPS> (reset)

CLIPS> (agenda)

0 all-students-passed: f-0,

For a total of 1 activation.

CLIPS>

След добавянето на фактс (student Bob), правилото вече не се удовлетворява, тъй като Bob не е преминал четене, писане и смятане.

CLIPS> (assert (student Bob))

<Fact-1>

CLIPS> (agenda)

CLIPS>

Правилото не се удовлетворява и след като Bob е преминал четене и писане, тъй като той все още не е преминал смятане.

CLIPS> (assert (reading Bob) (writing Bob))

<Fact-3>

CLIPS> (agenda)

CLIPS>

След като Bob е минал смятане, правилото all-students-passed се реактивира

CLIPS> (assert (arithmetic Bob))

<Fact-4>

CLIPS> (agenda)

0 all-students-passed: f-0,

For a total of 1 activation.

CLIPS>

Ако се добави нов ученик, правилото се премахва от конфликтното множество, тъй като John не е преминал четене, писане и смятане.

CLIPS> (assert (student John))

<Fact-5>

CLIPS> (agenda)

CLIPS>

Премахването на фактите за двамата студенти отново реактивира правилото.

CLIPS> (retract 1 5)

CLIPS> (agenda)

0 all-students-passed: f-0,

For a total of 1 activation.

CLIPS>

Важна забележка

Елементът forall е реализиран, чрез елемент not. Тъй като елементът на условието not може да предизвика добавяне на образец на факт (initial-fact) или образец на инстанция (instance-object), трябва да се изпълни команда reset.

Елемент на условието logical

Елементът на условието logical дава възможност за поддръжка на истинността на удовлетворяващи обекти (факти или инстанции) създадени в правилата които използват елемента logical. Един удовлетворяващ обект създаден в дясната част (или като резултат от действия изпълнени в дясната част) може да бъде бъде логически зявисим от удовлетворяващите обекти които са удовлетворили образеца обхванат от елемента logical в лявата част на правилото. Удовлетворяващите обекти в logical осигуряват логическа поддръжка на фактите и инстанциите създадени в RHS. на правилото. Един удовлетворяващ обект може да бъде логически поддържан от повече от една група от удовлетворяващи образци от същото или различни правила. Ако кой да е от подържащите обекти се премахне от групата поддържащи обекти (и няма други поддържащи групи), то удовлетворяващия обект се премахва.

Ако един удовлетворяващ обект се създава без логическа поддръжка (напр. чрез deffacts, definstances, като резултат от команда или от правило без логически елемент), то удовлетворяващия обект има безусловна поддръжка. Тази безусловна поддръжка премахва всички логически поддръжки за обекта (без да премахва самия обект). Всяка добавена логическа поддръжка за безусловен обект се игнорира. Премахването на правило което е генерирало логическа поддръжка на обект, премахва логическата поддръжка генерирана от правилото (но не предизвиква премахване на обекта ако не е останала логическа поддръжка за него).

Синтаксис

<logical-CE> ::= (logical <conditional-element>+)

Елементът на условието logical групира обекти заедно, точно както елементът and. Той може да се използва в конюнкция с and, or и not елементи. Само първите N обекта на едно правило обаче могат да имат елемент logical. Например както в следното правило:

(defrule ok

(logical (a))

(logical (b))

(c)

=>

(assert (d)))

Следните правила не са правилни:

(defrule not-ok-1

(logical (a))

(b)

(logical (c))

=>

(assert (d)))

(defrule not-ok-2

(a)

(logical (b))

(logical (c))

=>

(assert (d)))

(defrule not-ok-3

(or (a)

(logical (b)))

(logical (c))

=>

(assert (d)))

Пример:

CLIPS> (clear)

CLIPS>

(defrule rule1

(logical (a))

(logical (b))

(c)

=>

(assert (g) (h)))

CLIPS>

(defrule rule2

(logical (d))

(logical (e))

(f)

=>

(assert (g) (h)))

CLIPS>

Следните команди илюстрират как работи logical:

CLIPS> (watch facts)

CLIPS> (watch activations)

CLIPS. (watch rules)

CLIPS> (assert (a) (b) (c) (d) (e) (f))

==> f-0 (a)

==> f-1 (b)

==> f-2 (c)

==> Activation 0 rule1: f-0,f-1,f-2

==> f-3 (d)

==> f-4 (e)

==> f-5 (f)

==> Activation 0 rule2: f-3,f-3,f-5

<Fact-5>

CLIPS> (run)

FIRE 1 rule2: f-3,f-4,f-5 ; 1st rule adds logical support

==> f-6 (g)

==> f-7 (h)

FIRE 2 rule1: f-0,f-1,f-2 ; 2nd rule adds further support

CLIPS> (retract 1)

<== f-0 (a) ; Removes 1st support for (g) and (h)

CLIPS> (assert (h)) ; (h) is unconditionally supported

FALSE

CLIPS> (retract 3)

<== f-3 (d) ; Removes 2nd support for (g)

<== f-6 (g) ; (g) has no more support

CLIPS> (unwatch all)

CLIPS>

Елементът logical може да се използва с обект за създаване на обекти, които са логически зависими от промени на определени слотове в удовлетворяващата инстанция, а не на всички слотове. Това не може да се постигне с темплейтни факти, тъй като промяната на слот на темплейт всъчщност включва премахване на стария темплейтен факт и добавяне на нов, докато промяната на слот на инстанция се извършва без подмяна. Примерът по-долу илюстрира това поведение:

CLIPS> (clear)

CLIPS>

(defclass A (is-a USER)

(role concrete)

(pattern-match reactive)

(slot foo (create-accessor write))

(slot bar (create-accessor write)))

CLIPS>

(deftemplate A

(slot foo)

(slot bar))

CLIPS>

(defrule match-A-s

(logical (object (is-a A) (foo ?))

(A (foo ?)))

=>

(assert (new-fact)))

CLIPS> (make-instance a of A)

[a]

CLIPS> (assert (A))

<Fact-0>

CLIPS> (watch facts)

CLIPS> (run)

==> f-1 (new-fact)

CLIPS> (send [a] put-bar 100)

100

CLIPS> (agenda)

CLIPS> (modify 0 (bar 100))

<== f-0 (A (foo nil) (bar nil))

<== f-1 (new-fact)

==> f-2 (A (foo nil) (bar 100))

<Fact-2>

CLIPS> (agenda)

0 match-A-s: [a],f-2

For a total of 1 activation.

CLIPS> (run)

==> f-3 (new-fact)

CLIPS> (send [a] put-foo 100)

<== f-3 (new-fact)

100

CLIPS> (agenda)

0 match-A-s: [a],f-2

For a total of 1 activation.

CLIPS> (unwatch facts)

CLIPS>

Автоматично добавяне и преподреждане на LHS елементи на условие

При определени обстоятелства, CLIPS добавя допълнителни елементи на условието към правилата (обикновено за целите на алгоритъма на съпоставяне на CLIPS). Има два образеца поподразбиране използвани в CLIPS: initial-fact и initial-object. Образеца initial-fact е:

(initial-fact)

а initial-object:

(object (is-a INITIAL-OBJECT) (name [initial-object]))

Правила без елемент на условието в LHS

Към всяко правило което няма образци в лявата си част се добавя initial-fact (освен ако факта не е забранен, чрез съответното конфигуриране на CLIPS, като в този случай в лявата част се добавя (initial-object). Например следното правило:

(defrule example-1

=>

)

ще се промени по следния начин:

(defrule example-1

(initial-fact)

=>

)

test и not като първи елемент на and

test и not елементи на условието използвани като първи елементи в and елемент имат initial-fact или initial-object образец добавен непосредствено след тях. Ако първия образец предшестващ съответния елемент елемента е образец на факт, се добавя initial-fact. Ако първия образец предшестващ съответния елемент елемента е образец на обект, се добавя initial-object. Ако няма предшестващ образец като елемент на условието, тъпа на образеца се определя, чрез следващия образец използвайки същата методология. Ако няма никакви образци в правилото, то преди съответния елемент на условието се поставя образеца initial-fact (освен ако фактите не са забранени, посредством съответно конфигуриране на CLIPS, като в този случай преди елемента на условието се добавя (initial-object). Например следните правила

(defrule example-2

(test (> 80 (startup-value)))

=>)

(defrule example-3

(test (> 80 (startup-value)))

(object (is-a MACHINE))

=>)

(defrule example-4

(machine ?x)

(not (and (not (part ?x ?y))

(inventoried ?x)))

=>)

ще се променят по следния начин:

(defrule example-2

(initial-fact)

(test (> 80 (startup-value)))

=>)

(defrule example-3

(object (is-a INITIAL-OBJECT) (name [initial-object]))

(test (> 80 (startup-value)))

(object (is-a MACHINE))

=>)

(defrule example-4

(machine ?x)

(not (and (initial-fact)

(not (part ?x ?y))

(inventoried ?x)))

=>)

test използван след not

Елементи на условието test, които непосредствено следват not се преместват автоматично от CLIPS след първия елемент на условието, който предшества not. Например следното правило:

(defrule example

(a ?x)

(not (b ?x))

(test (> ?x 5))

=>)

ще се промени по следния начин:

(defrule example

(a ?x)

(test (> ?x 5))

(not (b ?x))

=>)

Елемент or след not

Ако or непосредствено следва not, то not/or комбинацията се премахва с and/not комбинация, при което всеки елемент на условието съдържащ се в оригиналния or се обхваща от not и след това всички not се обхващат в единичен and. Например следното правило

(defrule example

(a ?x)

(not (or (b ?x) (c ?x)))

=>)

ще се промени по следния начин:

(defrule example

(a ?x)

(and (not (b ?x)) (not (c ?x)))

=>)

Бележки относно добавяне и преподреждане на образци

Има няколко въпроса относно добавянето и преподреждане на образци на които трябва да се обърне внимание:

1) Цялата лява част на едно правило се разглежда като намираща се в неявен and елемент;

2) Преобразуването на forall и exists елементи към еквивалентни not и and елементи се извършва преди да се добавят образци към лявата част на едно правило;

3) Най-общо, не е много добре да има test като първи елемент в and;

4) Изходът на команди като matches командата извеждат информация за елементи на условието които всъщност се добавят към LHS на правило, тъй като преподреждането и неявните добавяния могатда не съответстват точно на дефинираните от потребителя правила.

===========================================================================

Задачи за самостоятелна работа:

1. Да се напише темплейт за студенти с техни данни (име, ф.н., специалност, брой взети изпити, и др.) Да се определи автомаично в кой курс е студентът на базата на броя на взетите от него изпити:

0-10 -> I курс

11-20 -> II курс

21-30 -> II курс

31-40 -> IV курс

